
Abstract. This document contains some lecture notes for a seminar
on Temporal Equilibrium Logic (TEL) and its application to Answer
Set Programming (ASP) inside the 17th Reasoning Web Summer School
(RW 2021). TEL is a temporal extension of ASP that introduces tem-
poral modal operators as those from Linear-Time Temporal Logic. We
present the basic definitions and intuitions for Equilibrium Logic and
then extend these notions to the temporal case. We also introduce sev-
eral examples using the temporal ASP tool telingo.

1 Introduction

Answer Set Programming [4] (ASP) is nowadays one of the most successful
paradigms for declarative problem solving and practical Knowledge Representa-
tion. Based on the answer set (or stable model) semantics [17] for logic programs,
ASP constitutes a declarative formalism and a natural choice for solving static
combinatorial problems up to NP complexity (or ΣP

2 in the disjunctive case),
but has also been applied to problems that involve a dynamic component and a
higher complexity, like planning, well-known to be PSPACE-complete [5]. The
use of ASP for temporal scenarios has been frequent since its early application
for reasoning about actions and change [18]. Commonly, dynamic scenarios in
ASP deal with transition systems and discrete time: instants are represented as
integer values for a time-point parameter, added to all dynamic predicates. This
temporal parameter is bound to a finite interval, from 0 to a maximum time-
step n (usually called the horizon). Problems involving temporal search, such as
planning or temporal explanation, are solved by multiple calls to the ASP solver
and gradually increasing the horizon length.

Although this methodology is simple and provides a high degree of flexibility,
it lacks for a differentiated treatment of temporal expressions (the time param-
eter is just one more logical variable to be grounded) and prevents the reuse of

the large corpora of techniques and results well-known from the temporal logic
literature. In an attempt to overcome these limitations, the approach called Tem-
poral Equilibrium Logic [7, 2] introduced a logical formalisation that combines
ASP with modal temporal operators. This formalism constitutes an extension of
Equilibrium Logic [21] which, in its turn, is a complete logical characterisation
of (standard) ASP based on the intermediate logic of Here-and-There (HT) [20].
As a result, TEL is an expressive non-monotonic modal logic that shares the
syntax of Linear-Time Temporal Logic (LTL) [22] but interprets temporal for-
mulas under a non-monotonic semantics that properly extends stable models.
This semantics is based on the idea of selecting some LTL temporal models of
a theory Γ that satisfy some minimality condition, when examined under the
weaker logic of temporal HT (THT). Thus, a temporal stable model of Γ is a
kind of selected LTL model of Γ , and so, it has the form of a sequence of states,
usually called a trace.

In the rest of this document we will first introduce some intuitions about
Equilibrium Logic from a rule-based reasoning perspective and then shift to the
definition of its temporal extension. After that, we will provide several examples
and talk about their practical implementation in the ASP tool called telingo.
Finally, we will close with some conclusions and open topics for future work or
currently under study.

2 Rule-based reasoning and Equilibrium Logic

In this section, we partly reproduce the motivations included in [12] (see that
paper for further detail). ASP is a rule-based paradigm sharing the same syntax
as the logic programming language Prolog but with a different reading. Take a
rule of the form:

smoke :- fire. (1)

ASP uses a bottom-up reading of (1): “smoke is produced by fire.” That is,
whenever fire belongs to our current set of beliefs or certain facts, smoke must
also be included in that set too. On the contrary, Prolog’s top-down reading
could be informally stated as “to obtain smoke, we need fire.” That is, the
rule describes a procedure to get smoke as a goal which consists in pursuing
fire as a new goal. Regardless of the application direction, it seems clear that
rules have a conditional form with a right-hand condition (body) and a left-hand
consequent (head) that in our example (1) respectively correspond to fire and
smoke. Thus, a straightforward logical formalisation would be understanding (1)
as the implication fire → smoke in classical propositional logic. This guarantees,
for instance, that if we add fire as a program fact, we will get smoke as a
conclusion (by application of modus ponens). So, the “operational” aspect of
rule (1) can be captured by classical implication. However, the semantics of a
classical implication is not enough to cover the intuitive meaning of a program
rule. If our program only contains (1) and we read it as a rule, it is clear that fire
is not satisfied, since no rule can yield that atom, and so, smoke is not obtained

either. However, implication fire → smoke, which amounts to the classically
equivalent disjunction ¬fire ∨ smoke, has three classical models: ∅ (both atoms
false), {smoke} and {fire, smoke}. Note that the two last models seem to consider
situations in which smoke or fire could be arbitrarily assumed as true, even
though the program provides no way to prove them. An important observation
is that ∅ happens to be the smallest model (with respect to set inclusion). This
model is interesting because, somehow, it reflects the principle of not adding
arbitrary true atoms that we are not forced to believe, and it coincides with the
expected meaning for a program just containing (1). The existence of a least
classical model is, in fact, guaranteed for logic programs without negation (or
disjunction), so-called positive logic programs, and so, it was adopted as the main
semantics [14] for logic programming until the introduction of negation. However,
when negation came into play, classical logic was revealed to be insufficient again,
even under the premise of minimal models selection. Suppose we have a program
Π1 consisting of the rules:

fill :- empty, not fire. (2)

empty. (3)

where (2) means that we always fill our gas tank if it is empty and there is
no evidence on fire, and (3) says that the tank is empty indeed. As before,
fire cannot be proved (it is not head of any rule) and so, the condition of (2)
is satisfied, producing fill as a result. The straightforward logical translation
of (2) is empty ∧ ¬fire → fill that, in combination with fact (3), produces three
models: T1 = {empty ,fill}, T2 = {empty ,fire} and T3 = {empty ,fire,fill}. Un-
fortunately, there is no least classical model any more: both T1 (the expected
model) and T2 are minimal with respect to set inclusion. After all, the previous
implication is classically equivalent to empty → fire∨fill which does not capture
the directional behaviour of rule (2). The undesired minimal model T2 is assum-
ing fire to be true, although there is no way to prove that fact in the program.
So, apparently, classical logic is too weak for capturing the meaning of logic
programs in the sense that it provides the expected model(s), but also accepts
other models (like T2 and T3) in which some atoms are abitrarily assumed to be
true but not “justified by the program.”

Suppose we had a way to classify true atoms distinguishing between those
just being an assumption (classical model T ) and those being also justified or
proved by program rules. In our intended models, the set of justified atoms should
precisely coincide with the set of assumed ones in T . As an example, suppose our
assumed atoms are T3 = {empty ,fire, smoke}. Any justification should include
empty because of fact (3). However, rule (2) seems to be unapplicable, because
we are currently assuming that fire is possibly true, fire ∈ T3, and so ‘not
fire’ is not acceptable – there is some (weak) evidence about fire. As a result,
atom fill is not necessarily justified and we can only derive {empty}, which is
strictly smaller than our initial assumption T3. Something similar happens for
assumption T2 = {empty ,fire}. If we take classical model T1 = {empty ,fill}
instead as an initial assumption, then the body of rule (2) becomes applicable,

since no evidence on fire can be found, that is, fire 6∈ T1. As a result, the justified
atoms are now {empty ,fill} = T1 and the classical model T1 becomes the unique
intended (stable) model of the program.

The method we have just used with the example can be seen as an informal
description of the original definition of the stable models semantics [17]. This
definition consisted of classical logic reinforced with an extra-logical program
transformation (for interpreting negation) and then using application of rules
to obtain the actually derived or justified information. We show next how it
is possible to provide an equivalent definition that exclusively resorts to logical
concepts but using a different underlying formalism, weaker than classical logic.

Although, as we have seen, our interest is focused on rules, the semantics of
Equilibrium Logic [21] can be defined on any arbitrary propositional formula.
Covering arbitrary formulas is, in fact, simpler and more homogeneous than the
original definition of stable models based on the reduct syntactic transformation.
Equilibrium models are defined by a models selection criterion on top of the
intermediate logic of Here-and-There (HT) [20], stronger than intuitionistic logic
but weaker than classical logic. The latter can be seen as a three-valued logic
where an atom can be false, assumed or proved, as we discussed before. Formally,
an HT interpretation is a pair of sets of atoms 〈H,T 〉 satisfying H ⊆ T so that:
any atom p ∈ H is considered as proved or justified ; an atom p ∈ T is considered
as assumed ; and any atom p 6∈ T is understood as false. As we can see, proved
implies assumed, that is, the set of justified atoms H is always a subset of the
assumed ones T . Intuitively, T acts as our “initial assumption” while the subset
H contains those atoms from T currently considered as justified. Let At be the
collection of all atomic formulas in our given language. ThenH ⊆ T ⊆ At and all
atoms in At \T are considered false in this model. An HT interpretation 〈H,T 〉
is said to be total when H = T (that is, when all assumptions are justified).

As we did for atoms, formulas can also be considered to be false, assumed
or proved. We will use a satisfaction relation 〈H,T 〉 |= ϕ to represent that
〈H,T 〉 makes formula ϕ to be proved or justified. Sometimes, however, we may
happen that this relation does not hold 〈H,T 〉 6|= ϕ while in classical logic
satisfaction T |= ϕ using the assumptions in T is true. Then, we may say that
the formula is just assumed. Finally, when ϕ is not even classically satisfied by
T , T 6|= ϕ, we can guarantee that the formula is false. Formally, the fact that
an interpretation 〈H,T 〉 satisfies a formula ϕ (or makes it justified), written
〈H,T 〉 |= ϕ, is recursively defined as follows:

– 〈H,T 〉 6|= ⊥
– 〈H,T 〉 |= p iff p ∈ H
– 〈H,T 〉 |= ϕ ∧ ψ iff 〈H,T 〉 |= ϕ and 〈H,T 〉 |= ψ
– 〈H,T 〉 |= ϕ ∨ ψ iff 〈H,T 〉 |= ϕ or 〈H,T 〉 |= ψ
– 〈H,T 〉 |= ϕ → ψ iff both (i) T |= ϕ implies T |= ψ and (ii) 〈H,T 〉 |= ϕ

implies 〈H,T 〉 |= ψ

By abuse of notation, we use ‘|=’ both for classical and for HT-satisfaction: the
ambiguity is resolved by the form of the left interpretation (a single set T for
classical and a pair 〈H,T 〉 for HT). We say that an interpretation 〈H,T 〉 is a

model of a theory (set of formulas) Γ iff 〈H,T 〉 |= ϕ for all ϕ ∈ Γ . We say that
a propositional theory Γ entails some formula ϕ, written Γ |= ϕ, if any model
of Γ is also a model of ϕ.

As we can see, everything is pretty standard excepting for the interpretation
of implication, which imposes a stronger condition than in classical logic. In order
to satisfy 〈H,T 〉 |= ϕ → ψ, the standard condition would be (ii), that is, if the
antecedent holds, then the consequent must hold too. In our case, the reading is
closer to an application of modus ponens in an inference rule: if the antecedent
is proved, then we can also prove the consequent. This condition, however, is
further reinforced by (i) which informally means that our set of assumptions T
classically satisfy the implication ϕ→ ψ as well.

The following proposition tells us that satisfaction for total models amounts
to classical satisfaction:

Proposition 1. For any formula ϕ and set of atoms T , 〈T, T 〉 |= ϕ iff T |= ϕ
in classical logic.

We may read this result saying that classical models are a subset of HT models
(they correspond to total HT models). This immediately means that any HT
tautology is also a classical tautology. The opposite does not hold, namely, there
are classical tautologies that are not HT tautologies. We will see later several
examples.

Classical satisfaction for T allows us to keep the three-valued reading (false,
assumed or proved) also for formulas in the following way. We say that 〈H,T 〉
makes formula ϕ:

– proved when 〈H,T 〉 |= ϕ,
– assumed when T |= ϕ,
– false when T 6|= ϕ.

Interestingly, as happened with atoms, formulas also satisfy that anything proved
must also be assumed. This is stated as the following property called persistence:

Proposition 2 (Persistence). For any formula ϕ and any HT interpretation
〈H,T 〉 we can show that, if ϕ is proved then it is also assumed, that is: 〈H,T 〉 |=
ϕ implies T |= ϕ.

Notice that we did not provide satisfaction of negation ¬ϕ. This is because
negation is not included above because it can be defined in terms of implication
as the formula ϕ→ ⊥, as happens in intuitionistic logic. Using that abbreviation
and after some analysis, it can be proved that 〈H,T 〉 |= ¬ϕ amounts to T 6|= ϕ,
that is, ¬ϕ is justified simply when ϕ is not assumed, that is, when it is false.

Apart from negation, we also define the common Boolean operators > def
= ¬⊥

and ϕ↔ ψ
def
= (ϕ→ ψ) ∧ (ψ → ϕ).

If we apply the persistence property to our example program Π1, this means
that any of its models 〈H,T 〉 |= Π must satisfy T |= Π as well. As we saw, we
have only three possibilities for the latter, T1, T2 and T3. On the other hand, the
program fact (3) fixes empty ∈ H. Now, take assumption T1 = {empty ,fill}. The

only model we get is 〈T1, T1〉 because the other possible subset H = {empty}
of T1 does not satisfy (2): empty is justified, fire is false, so we should get
fill . Take T2 instead. Apart from 〈T2, T2〉, in this case we also get a model
〈H,T2〉 with H = {empty}. In such a case, fire is only assumed true, but not
proved. As a result, the rule is satisfied because its condition ¬fire is false (we
have some evidence on fire) and so 〈{empty}, T2〉 becomes a model. This is a
clear evidence that our initial assumption adding fire is not necessarily proved
when we check the program rules. In the case of T3 = {empty ,fire,fill} we have
a similar situation. Interpretations with H = {empty}, H = {empty ,fire} or
H = {empty ,fill} are also models. Note that in all of them, the only atom that
is always proved is empty , pointing out again that fire or fill are not necessarily
justified (cannot be proved using the program rules).

It must be understood that, at this point, the tag “justified” or “proved” just
refers to a second kind of truth, stronger than “assumed.” This tag will only ac-
quire a real “provability” meaning once we introduce a models minimisation. For
instance, for the formula fire → smoke, we will have a model H = T = {smoke}
where smoke is being considered justified. We still miss some minimisation cri-
terion to consider justified or proved only those atoms and formulas that we are
certain to be so. This minimisation selects some particular HT models and will
follow the intuitive idea: given a fixed set of assumptions T , minimise proved
atoms H. If we additionally require that anything assumed must be eventually
proved, we get the following definition of equilibrium models first introduced by
Pearce [21]:

Definition 1 (Equilibrium Model). A total HT interpretation 〈T, T 〉 is an
equilibrium model of a theory Γ if 〈T, T 〉 |= Γ and there is no H ⊂ T such
that 〈H,T 〉 |= Γ . When this happens, we also say that T is a stable model (or
answer set) of Γ .

From the logical point of view, it is now an easy task to define a stable model.
The intuition is that we will be interested in cases where anything assumed true
in set T eventually becomes necessarily proved, ie , H = T is the only possibility
for assumption T .

Back to our example, the only stable model of Π1 is the expected T1 =
{empty ,fill}. This is because for the other two classical models T2 = {empty ,fire}
and T3 = {empty ,fill ,fire} we could see in the previous section that there were
smaller sets H ′ that formed possible models of the program such as 〈{empty}, T2〉
or 〈{empty}, T3〉. In the case of T1, however, the only obtained model is 〈T1, T1〉
and no smaller H ⊂ T1 can be used to form a model.

By selecting the equilibrium models, we obtain a non-monotonic entailment
relation, that is, we may obtain conclusions that, after we add new information,
can be retracted. For instance, in our example, if we are now said that fire
has been observed and we take program Π2 = Π1 ∪ {fire}, the classical models
become T4 = {empty ,fire} and T5 = {empty ,fire,fill}. Clearly, T4 will be a
stable model, since there is no way to remove any of the two atoms empty , fire
that are now facts in program Π2. However, T5 is not in equilibrium, since we can

form H = {empty ,fire} and 〈H,T5〉 |= Π2 because rule (2) is always satisfied,
as its body is falsified since T5 6|= ¬fire. From the only stable model we obtain,
we conclude that we cannot fill the tank any more, while this atom was among
the previous conclusions when we had no information about fire.

As said before, some classical tautologies are not HT tautologies. An inter-
esting example is the law of excluded middle a ∨ ¬a. Intuitively, this formula
means that a has to be proved or assumed and, accordingly, it has the HT coun-
termodel 〈∅, {a}〉. Therefore, if we include a ∨ ¬a for some atom in our theory,
we force that assuming a is enough to consider it proved, so p will somehow
behave “classically.” Moreover, if we add the formula a ∨ ¬a for all atoms in
the signature At , then all models are forced to be total 〈T, T 〉 and Equilibrium
Logic collapses into classical logic.

3 Temporal Equilibrium Logic

An important advantage of the definition of stable models based on Equilibrium
Logic is that it provides a purely logical characterisation that has no syntactic
limitations (it applies to arbitrary propositional formulas) and is easy to extend
with the incorporation of new constructs or the definition of new combined logics.
As happened with Equilibrium Logic, the definition of (Linear-time) Temporal
Equilibrium Logic (TEL) is done in two steps. First, we define a monotonic tem-
poral extension of HT, called (Linear-time) Temporal Here-and-There (THT)
and, second, we select some models from THT that are said to be in equilib-
rium, obtaining in this way a non-monotonic entailment relation.

We reproduce next part of the contents from [1] and [10]. The original defini-
tion of TEL was thought as a direct non-monotonic extension of standard LTL,
so that models had the form of infinite traces. However, this rules out computa-
tion by ASP technology and is unnatural for applications like planning, where
plans amount to finite prefixes of one or more traces [13]. In a recent line of
research [11], TEL was extended to cope with finite traces (which are closer to
ASP computation). On the one hand, this amounts to a restriction of THT and
TEL to finite traces. On the other hand, this is similar to the restriction of LTL
to LTLf advocated by [13]. Our new approach, dubbed TELf , has the following
advantages. First, it is readily implementable via ASP technology. Second, it can
be reduced to a normal form which is close to logic programs and much simpler
than the one obtained for TEL. Finally, its temporal models are finite and offer
a one-to-one correspondence to plans. Interestingly, TELf also sheds light on
concepts and methodology used in incremental ASP solving when understand-
ing incremental parameters as time points. Another distinctive feature of TELf
is the inclusion of future as well as past temporal operators. When using the
causal reading of program rules, it is generally more natural to draw upon the
past in rule bodies and to refer to the future in rule heads. As well, past oper-
ators are much easier handled computationally than their future counterparts
when it comes to incremental reasoning, since they refer to already computed
knowledge.

In what follows, we present the general logics THT (monotonic) and TEL
(non-monotonic) allowing traces of any length (possibly infinite), and will later
on use the subindices ω or f to denote the particular cases where traces are
always infinite or always finite, respectively. The syntax of THT (and TEL) is
the same as for LTL with past operators. Given a (countable, possibly infinite)
set At of propositional variables (called alphabet), temporal formulas ϕ are
defined by the grammar:

ϕ ::= a | ⊥ | ϕ1 ⊗ ϕ2 | •ϕ | ϕ1 S ϕ2 | ϕ1 T ϕ2 | ◦ϕ | ϕ1 U ϕ2 | ϕ1 R ϕ2 | ϕ1 W ϕ2

where a ∈ At is an atom and ⊗ is any binary Boolean connective ⊗ ∈ {→,∧,∨}.
The last six cases correspond to the temporal connectives whose names are listed
below:

Past • for previous
S for since
T for trigger

Future ◦ for next
U for until
R for release
W for while

We also define several common derived temporal operators:

�ϕ
def
= ⊥ T ϕ always before

�ϕ
def
= > S ϕ eventually before

I
def
= ¬•> initial

•̂ϕ def
= •ϕ ∨ I weak previous

�ϕ
def
= ⊥ R ϕ always afterward

3ϕ def
= >U ϕ eventually afterward

F
def
= ¬◦> final

◦̂ϕ def
= ◦ϕ ∨ F weak next

A (temporal) theory is a (possibly infinite) set of temporal formulas. Note that we
use solid operators to refer to the past, while future-time operators are denoted
by outlined symbols.

As happens with HT with respect to classical logic, logics THT and LTL
share the same syntax but, they have a different semantics, the former being a
weaker logic than the latter. The semantics of LTL relies on the concept of a
trace, a (possibly infinite) sequence of states, each of which is a set of atoms.
For defining traces, we start by introducing some notation to deal with intervals
of integer time points. Given a ∈ N and b ∈ N ∪ {ω}, we let [a..b] stand for
the set {i ∈ N | a ≤ i ≤ b}, [a..b) for {i ∈ N | a ≤ i < b} and (a..b] for
{i ∈ N | a < i ≤ b}. In LTL, a trace T of length λ over alphabet At is
a sequence T = (Ti)i∈[0..λ) of sets Ti ⊆ At . We sometimes use the notation

|T| def= λ to stand for the length of the trace. We say that T is infinite if |T| = ω
and finite if |T| ∈ N. To represent a given trace, we write a sequence of sets
of atoms concatenated with ‘·’. For instance, the finite trace {a} · ∅ · {a} · ∅ has
length 4 and makes a true at even time points and false at odd ones. For infinite
traces, we sometimes use ω-regular expressions like, for instance, in the infinite
trace ({a} · ∅)ω where all even positions make a true and all odd positions make
it false.

A state i is represented as a pair of sets of atoms 〈Hi, Ti〉 with Hi ⊆ Ti ⊆ At
where Hi (standing for “here”) contains the proved atoms, whereas Ti (standing

for “there”) contains the assumed atoms. On the other hand, false atoms are just
the ones not assumed, captured by At \Ti. An HT-trace of length λ over alphabet
At is a sequence of pairs (〈Hi, Ti〉)i∈[0..λ) with Hi ⊆ Ti for any i ∈ [0..λ). For
convenience, we usually represent the HT-trace as the pair 〈H,T〉 of traces
H = (Hi)i∈[0..λ) and T = (Ti)i∈[0..λ). Given M = 〈H,T〉, we also denote its

length as |M| def= |H| = |T| = λ. Note that the two traces H, T must satisfy a
kind of order relation, since Hi ⊆ Ti for each time point i. Formally, we define
the ordering H ≤ T between two traces of the same length λ as Hi ⊆ Ti for each
i ∈ [0..λ). Furthermore, we define H < T as both H ≤ T and H 6= T. Thus, an
HT-trace can also be defined as any pair 〈H,T〉 of traces such that H ≤ T. The
particular type of HT-traces satisfying H = T are called total.

Given any HT-trace M = 〈H,T〉, we define the THT satisfaction of formulas
as follows.

Definition 2 (THT-satisfaction). An HT-trace M = 〈H,T〉 of length λ over
alphabet At satisfies a temporal formula ϕ at time point k ∈ [0..λ), written
M, k |= ϕ, if the following conditions hold:

1. M, k |= > and M, k 6|= ⊥
2. M, k |= a if a ∈ Hk for any atom a ∈ At
3. M, k |= ϕ ∧ ψ iff M, k |= ϕ and M, k |= ψ
4. M, k |= ϕ ∨ ψ iff M, k |= ϕ or M, k |= ψ
5. M, k |= ϕ→ ψ iff 〈H′,T〉, k 6|= ϕ or 〈H′,T〉, k |= ψ, for all H′ ∈ {H,T}
6. M, k |= •ϕ iff k > 0 and M, k−1 |= ϕ
7. M, k |= ϕ S ψ iff for some j ∈ [0..k], we have M, j |= ψ and M, i |= ϕ for

all i ∈ (j..k]
8. M, k |= ϕ T ψ iff for all j ∈ [0..k], we have M, j |= ψ or M, i |= ϕ for some

i ∈ (j..k]
9. M, k |= ◦ϕ iff k + 1 < λ and M, k+1 |= ϕ

10. M, k |= ϕ U ψ iff for some j ∈ [k..λ), we have M, j |= ψ and M, i |= ϕ for
all i ∈ [k..j)

11. M, k |= ϕR ψ iff for all j ∈ [k..λ), we have M, j |= ψ or M, i |= ϕ for some
i ∈ [k..j)

12. M, k |= ϕWψ iff for all j ∈ [k..λ), we have 〈H′,T〉, j |= ϕ or 〈H′,T〉, i 6|= ψ
for some i ∈ [k..j) and for all H′ ∈ {H,T}

ut

In general, these conditions inherit the interpretation of connectives from
LTL (with past operators) with just a few differences. A first minor variation
is that we allow traces of arbitrary length λ, including both infinite (λ = ω)
and finite (λ ∈ N) traces. A second difference with respect to LTL is the new
connective ϕWψ which is also a kind of temporally-iterated HT implication. Its
intuitive reading is “keep doing ϕ while condition ψ holds.” In LTL, ϕWψ would
just amount to ¬ψ R ϕ, but under HT semantics both formulas have a different
meaning, as the latter may provide evidence for ϕ even though the condition ψ
does not hold.

An HT-trace M is a model of a temporal theory Γ if M, 0 |= ϕ for all ϕ ∈ Γ .
We write THT (Γ, λ) to stand for the set of THT-models of length λ of a theory

Γ , and define THT (Γ )
def
= THT (Γ, ω)∪

⋃
λ∈N THT (Γ, λ). That is, THT (Γ ) is the

whole set of models of Γ of any length. For Γ = {ϕ}, we just write THT (ϕ, λ)
and THT (ϕ). We can analogously define LTL(Γ, λ), that is, the set of traces of
length λ that satisfy theory Γ , and LTL(Γ ), that is, the LTL-models of Γ any
length. We omit specifying LTL satisfaction since it coincides with THT when
HT-traces are total.

Proposition 3 ([2, 11]). Let T be a trace of length λ, ϕ a temporal formula,
and k ∈ [0..λ) a time point.

Then, T, k |= ϕ in LTL iff 〈T,T〉, k |= ϕ. ut

In fact, total models can be forced by adding the following set of excluded middle
axioms:

�(a ∨ ¬a) for each atom a ∈ At in the signature. (EM)

Proposition 4 ([2, 11]). Let 〈H,T〉 be an HT-trace and (EM) the theory con-
taining all excluded middle axioms for every atom a ∈ At . Then, 〈H,T〉 is a
model of (EM) iff H = T. ut

Satisfaction of derived operators can be easily deduced, as shown next.

Proposition 5 ([2, 11]). Let M = 〈H,T〉 be an HT-trace of length λ over
At . Given the respective definitions of derived operators, we get the following
satisfaction conditions:

12. M, k |= I iff k = 0
13. M, k |= •̂ϕ iff k = 0 or M, k−1 |= ϕ
14. M, k |= �ϕ iff M, i |= ϕ for some i ∈ [0..k]
15. M, k |= �ϕ iff M, i |= ϕ for all i ∈ [0..k]
16. M, k |= F iff k + 1 = λ
17. M, k |= ◦̂ϕ iff k + 1 = λ or M, k+1 |= ϕ
18. M, k |= 3ϕ iff M, i |= ϕ for some i ∈ [k..λ)
19. M, k |= �ϕ iff M, i |= ϕ for all i ∈ [k..λ)

ut

Given a set of THT-models, we define the ones in equilibrium as follows.

Definition 3 (Temporal Equilibrium/Stable Model). Let S be some set
of HT-traces. A total HT-trace 〈T,T〉 ∈ S is a temporal equilibrium model of
S iff there is no other H < T such that 〈H,T〉 ∈ S. The trace T is called a
temporal stable model (TS-model) of S. ut

We further talk about temporal equilibrium or temporal stable models of a
theory Γ when S = THT (Γ ), respectively. Moreover, we write TEL(Γ, λ)
and TEL(Γ ) to stand for the temporal equilibrium models of THT (Γ, λ) and
THT (Γ ) respectively. We write TSM(Γ, λ) and TSM(Γ ) to stand for the corre-
sponding sets of TS-models. One interesting observation is that, since temporal

equilibrium models are total models 〈T,T〉, due to Proposition 3, we obtain
TSM(Γ, λ) ⊆ LTL(Γ, λ) that is, temporal stable models are a subset of LTL-
models.

Temporal Equilibrium Logic (TEL) is the (non-monotonic) logic induced by
temporal equilibrium models. We can also define the variants TELω and TELf by
applying the corresponding restriction to infinite and finite traces, respectively.

As an example of non-monotonicity, consider the formula

�(•loaded ∧ ¬unloaded → loaded) (4)

that corresponds to the inertia for loaded , together with the fact loaded , describ-
ing the initial state for that fluent. Without entering into too much detail, this
formula behaves as the logic program with the rules:

loaded (0).

loaded(T) :- loaded(T-1), not unloaded(T).

for any time point T>0. As expected, for some fixed λ, we get a unique temporal
stable model of the form {loaded}λ. This entails that loaded is always true, viz.
�loaded , as there is no reason for unloaded to become true. Note that in the
most general case of TEL, we actually get one stable model per each possible
λ, including λ = ω. Now, consider formula (4) along with loaded ∧ ◦◦unloaded
which amounts to adding the fact unloaded(2). As expected, for each λ, the only
temporal stable model now is T = {loaded} · {loaded} · {unloaded} · ∅α where
α can be ∗ or ω. Note that by making ◦◦unloaded true, we are also forcing
|T| ≥ 3, that is, there are no temporal stable models (nor even THT-models)
of length smaller than three. Thus, by adding the new information ◦◦unloaded
some conclusions that could be derived before, such as �loaded , are not derivable
any more.

As an example emphasizing the behavior of finite traces, take the formula

�(¬a→ ◦a) (5)

which can be seen as a program rule “a(T+1) :− not a(T)” for any natural
number T. As expected, temporal stable models make a false in even states and
true in odd ones. However, we cannot take finite traces making a false at the
final state λ− 1, since the rule would force ◦a and this implies the existence of
a successor state. As a result, the temporal stable models of this formula have
the form (∅ · {a})+ for finite traces in TELf , or the infinite trace (∅ · {a})ω in
TELω.

Another interesting example is the temporal formula

�(¬◦a→ a) ∧�(◦a→ a).

The corresponding rules “a(T) :− not a(T+1)” and “a(T) :− a(T+1)” have no
stable model [15] when grounded for all natural numbers T. This is because
there is no way to build a finite proof for any a(T), as it depends on infinitely
many next states to be evaluated. The same happens in TELω, that is, we get
no infinite temporal stable model. However in TELf , we can use the fact that

◦a is always false in the last state. Then, �(¬◦a→ a) supports a in that state
and therewith �(◦a→ a) inductively supports a everywhere.

As an example of a temporal expression not so close to logic programming,
consider the formula �3a, which is normally used in LTLω to assert that a
occurs infinitely often. As discussed by [13], if we assume finite traces, then the
formula collapses to �(F→ a) in LTLf , that is, a is true at the final state (and
either true or false everywhere else). The same behavior is obtained in THTω
and THTf , respectively. However, if we move to TEL, a truth minimization
is additionally required. As a result, in TELf , we obtain a unique temporal
stable model for each fixed λ ∈ N, in which a is true at the last state, and false
everywhere else. Unlike this, TELω yields no temporal stable model at all. This
is because for any T with an infinite number of a’s we can always take some
H from which we remove a at some state, and still have an infinite number of
a’s in H. Thus, for any total THTω-model 〈T,T〉 of �3a there always exists
some model 〈H,T〉 with strictly smaller H < T. Note that we can still specify
infinite traces with an infinite number of occurrences of a, but at the price of
removing the truth minimization for that atom. This can be done, for instance,
by adding the excluded middle axiom (EM) for atom a. In this way, infinite
traces satisfying �3a ∧ �(a ∨ ¬a) are those that contain an infinite number of
a’s. In fact, if we add the excluded middle axiom for all atoms, TEL collapses
into LTL, as stated below.

Proposition 6. Let Γ be a temporal theory over At and (EM) be the set of all
excluded middle axioms for all atoms in At .

Then, TSM( Γ ∪ (EM) ) = LTL(Γ ). ut

4 Computing Temporal Stable Models

Let us consider a more meaningful example, taking the Yale Shooting sce-
nario [19] where we must shoot a loaded gun to kill a turkey. A possible encoding
in TEL could be:

�(loaded ∧ ◦shoot → ◦dead) (6)

�(loaded ∧ ◦shoot → ◦unloaded) (7)

�(load → loaded) (8)

�(dead → ◦dead) (9)

�(loaded ∧ ¬◦unloaded → ◦loaded) (10)

�(unloaded ∧ ¬◦loaded → ◦unloaded) (11)

In this way, under TEL semantics, implication α → β has a similar behaviour
to a directional inference rule, normally reversed as β ← α or β :− α in logic
programming notation. The last two rules, (10)-(11), encode the inertia law for
fluents loaded and unloaded, respectively. Note the use of ¬ in these two rules:
it actually corresponds to default negation, that is, ¬α is read as “there is no



Temporal ASP: from logical foundations to practical use with telingo 13

evidence about α.” For instance, (10) is read as “if the gun was loaded and we
cannot prove that it will become unloaded then it stays loaded.”

Computation of temporal stable models is a complex task. THT-satisfiability
has been classified [8] as Pspace-complete, that is, the same complexity as
LTL-satisfiability, whereas TEL-satisfiability rises to ExpSpace-completeness,
as proved in [3]. In this way, we face a similar situation as in the non-temporal
case where HT-satisfiability is NP-complete like SAT, whereas existence of equi-
librium model (for arbitrary theories) is ΣP

2 -complete (like disjunctive ASP).
There exist a pair of tools, STeLP [6] and ABSTEM [9], that allow computing (infi-
nite) temporal stable models (represented as Büchi automata). These tools can
be used to check verification properties that are usual in LTL, like the typi-
cal safety, liveness and fairness conditions, but in the context of temporal ASP.
Moreover, they can also be applied for planning problems that involve an inde-
terminate or even infinite number of steps, such as the non-existence of a plan.
In most practical problems, however, we are normally interested in finite traces.
For that purpose, TELf is implemented in the telingo system, extending the
ASP system clingo to compute the temporal stable models of (non-ground)
temporal logic programs. To this end, it extends the full-fledged input language
of clingo with temporal operators and computes temporal models incrementally
by multi-shot solving using a modular translation into ASP. telingo is freely
available at github1. For instance, under telingo syntax, our theory (6)-(11)
would be represented2 as

#program dynamic.

dead :- shoot, ’loaded.

unloaded :- shoot, ’unloaded.

loaded :- load.

dead :- ’dead.

loaded :- ’loaded, not unloaded.

unloaded :- ’unloaded, not loaded.

The telingo input language actually allows the introduction of arbitrary
LTL formulas in constraints or past formulas in the rule bodies (conditions).
The syntax extends the full-fledged modeling language of clingo by the future
and past temporal operators listed in the first and fourth row of Table 1. To
support incremental ASP solving, telingo accepts a fragment of TELf called
past-future rules (see [11] for more details). A temporal formula is a past-future
rule if it has form Hd ← Bd where Bd and Hd are just temporal formulas
with the following restrictions: Bd and Hd contain no implications (other than
negations3), Bd contains no future operators, and Hd contains no past operators.

1 https://github.com/potassco/telingo
2 The left upper commas are read as previously and correspond to the past operator

dual of next ‘◦’. The � operator is implicit in all dynamic rules.
3 Recall that ¬ϕ def

= ϕ → ⊥ in the logic of here-and-there and thus in TELf , too.

&initial I initial &final F final
’p •p previous p’ ◦p next

< • previous > ◦ next
<? S since >? U until
<* T trigger >* R release
<? � eventually before >? 3 eventually afterward
<* � always before >* � always afterward

<: •̂ weak previous >: ◦̂ weak next

Table 1. Past and future temporal operators in telingo and TELf

An example of a past-future rule is, for instance, the formula

�(shoot ∧ •�shoot ∧�unloaded → 3fail) (12)

expressing the sentence: “If we shoot twice with a gun that was never loaded, it
will eventually fail.” The past-future fragment is not only quite expressive but
also rather natural when using the causal reading of program rules by drawing
upon the past in rule bodies and referring to the future in rule heads. Considering
that, past-future rules also serve as the design guideline for telingo’s input
language.

To this end, telingo allows for enclosing a nested temporal formula ϕ in
an expression of the form &tel{ϕ}. Formulas like ϕ are formed via the tempo-
ral operators in Line 3 to 8 in Table 1 along with the Boolean operators &, |,
~ for conjunction, disjunction, and negation, respectively (thus avoiding nested
implications). The underlying idea is to use the smaller symbol < as the basis
of all past operators, and to combine it with a question mark ? or a Kleene
star * depending on whether the semantics of the respective operator relies on
an existential or universal quantification over states. This is nicely exemplified
by the always and eventually operators, represented by <* and <?. In fact, the
symbols <* and <? are overloaded due to their usage as binary and unary opera-
tors. For a simple example, consider the formula •p∨�r represented as ‘&tel{<
p | <? p}’. Similarly, future operators are built with the greater symbol ‘>’
as their basis. More generally, temporal expressions of the form &tel{ϕ} are
treated like atoms in telingo’s input language (and constitute theory atoms in
clingo [?]); they are compiled away by telingo’s preprocessing that ultimately
yields present-centered logic programs. In order to keep this translation simple,
the current version of telingo, viz 2.1.1, restricts their occurrence in temporal
rules Hd ← Bd to being positive in Hd and preceded by one or two negations in
their body Bd .4 No restriction is imposed on their occurrences in integrity con-
straints. For example, the integrity constraint ‘shoot∧�unloaded∧•�shoot → ⊥’
is expressible in several alternative ways.

4 The extension to arbitrary occurrences is no hurdle and foreseen in future versions
of telingo.

:- &tel { shoot & <* unloaded & < <? shoot }.

:- shoot , &tel { <* unloaded & < <? shoot }.

:- shoot , &tel { <* unloaded }, &tel { < <? shoot }.

Alternatively, present-centered logic programs can be written directly by us-
ing the alternative notation for the common one-step operators • and ◦. Here, a
quote is used either at the beginning or the end of a predicate symbol to indicate
that the literal at hand must be true in the previous or next state in the trace,
respectively. For instance, •p(7) is represented by ’p(7), while ◦q(X) is q’(X).
For convenience, telingo 2.1.1 allows for using ◦ in singleton rule heads;5 as
above, this is compiled away during preprocessing.

The distinction between different types of temporal rules is done in telingo

via clingo’s #program directives [16], which allow us to partition programs into
subprograms. More precisely, each rule in telingo’s input language is associated
with a temporal rule r of form Hd ← Bd and interpreted as r, ◦̂�r, or �(F→ r)
depending on whether it occurs in the scope of a program declaration headed
by initial , dynamic, or final , respectively. Additionally, telingo offers always
for gathering rules preceded by � (thus dropping ◦̂ from dynamic rules). A rule
outside any such declaration is regarded to be in the scope of initial .

For illustration, we give in Listing 1.1 an exemplary telingo encoding of
the Fox, Goose and Beans Puzzle available at https://github.com/potassco/
telingo/tree/master/examples/river-crossing.

Once upon a time a farmer went to a market and purchased a fox, a
goose, and a bag of beans. On his way home, the farmer came to the bank
of a river and rented a boat. But crossing the river by boat, the farmer
could carry only himself and a single one of his purchases: the fox, the
goose, or the bag of beans. If left unattended together, the fox would eat
the goose, or the goose would eat the beans. The farmer’s challenge was
to carry himself and his purchases to the far bank of the river, leaving
each purchase intact. How did he do it?

(https://en.wikipedia.org/wiki/Fox,_goose_and_bag_of_beans_puzzle)

In Listing 1.1, lines 3-5 and 9-10 provide facts holding in all and the initial
states, respectively; this is indicated by the program directives headed by always

and initial. The dynamic rules in lines 14-22 describe the transition function.
The farmer moves at each time step (Line 14), and may take an item or not
(Line 15). Line 17 describes the effect of action move/1, Line 18 its precondition,
and Line 20 the law of inertia. The second part of the always rules give state
constraints in Line 24 and 25. The final rule in Line 29 gives the goal condition.

All in all, we obtain two shortest plans consisting of eight states in about 20
ms. Restricted to the move predicate, telingo reports the following solutions:

5 As above, the extension to disjunctions is no principal hurdle and foreseen in future
versions of telingo; currently they must be expressed by using &tel.

Listing 1.1. telingo encoding for the Fox, Goose and Beans Puzzle

#program always.

item(fox;beans;goose ).

route(river_bank ,far_bank ). route(far_bank ,river_bank ).

eats(fox ,goose ). eats(goose ,beans ).

#program initial.

at(farmer ,river_bank ).

at(X,river_bank) :- item(X).

#program dynamic.

move(farmer ).

0 { move(X) : item(X) } 1.

at(X,B) :- ’at(X,A), move(X), route(A,B).

:- move(X), item(X), ’at(farmer ,A), not ’at(X,A).

at(X,A) :- ’at(X,A), not move(X).

#program always.

:- at(X,A), at(X,B), A<B.

:- eats(X,Y), at(X,A), at(Y,A), not at(farmer ,A).

#program final.

:- at(X,river_bank ).

#show move /1.

#show at/2.

Time Solution 1 Solution 2

1
2 move(farmer) move(goose) move(farmer) move(goose)

3 move(farmer) move(farmer)

4 move(beans) move(farmer) move(farmer) move(fox)

5 move(farmer) move(goose) move(farmer) move(goose)

6 move(farmer) move(fox) move(beans) move(farmer)

7 move(farmer) move(farmer)

8 move(farmer) move(goose) move(farmer) move(goose)

We have chosen this example since it was also used by [6] to illustrate the
working of STeLP, a tool for temporal answer set programming with TELω. We
note that STeLP and telingo differ syntactically in describing transitions by
using next or previous operators, respectively. Since telingo extends clingo’s
input language, it offers a richer input language, as witnessed by the cardinality
constraints in Line 15 in Listing 1.1. Finally, STeLP uses a model checker and
outputs an automaton capturing all infinite traces while telingo returns finite
traces corresponding to plans.

As a second example, consider the following problem6 proposed by Professor
J. Moore from the University of Texas at Austin and submitted to the Texas
Action Group (TAG) discussion group.

Consider two processes, A and B, each of which is reading and writing a
shared variable C. Each process is in an infinite loop, repeatedly execut-
ing: C = C + C; By this we mean “read the value of C, read the value
of C again, add the two results and store the sum in C.” The two reads
store the values in “local registers” of the process. Reads and writes are
atomic but there is no synchronization between the two processes. The
initial value of C is 1. Problem: Given an arbitrary positive integer n is
there an execution that assigns C the value n?

A first possible encoding of this problem in telingo could look simply be the
one shown in Listing 1.2. In this encoding, action fetch(P) reflects the fact that
the CPU has non-deterministically decided to execute the next instruction from
process P, encoding the process interleaving in that way. This non-deterministic
choice is encoded in lines 23-26. The most significative feature of this encoding
is that we keep an auxiliary fluent i(P) to stand for the current instruction
pointerof each process P. An interesting observation is that, once we allow tem-
poral expressions in the rule bodies and constraints, we can sometimes replace
auxiliary fluents in favour of temporal queries about the past execution. In our
example, this leads to a second encoding shown in Listing 1.3. In this case, we
have removed the fluent capturing the instruction pointer and replaced it by the
constraints in lines 33-35. Lines 33-34 mean that if we fetch instruction I for
a process P already fetched in the past, then the last instruction fetched for P

(whenever this is located in the past) must be I+2 modulo 3 (that is, the previous
one in the cyclic order of instructions 0,1,2,0,. . . ). Line 35 forces that the first
instruction to be executed by any process P is 0: we cannot fetch instructions 1
or 2 if process P has not been already fetched.

5 Conclusions

These recent results open several interesting topics for future study. First, it
would be interesting to adapt existing model checking techniques (based on
automata construction) for temporal logics to solve the problem of existence

6 https://www.cs.utexas.edu/users/vl/tag/jmoore_discussion

of temporal stable models. This was done for infinite traces in [8, 6], but no
similar method has been implemented for finite traces on TELf . The importance
of having an efficient implementation of such a method is that it would allow
deciding non-existence of a plan in a given planning problem, something not
possible by current incremental solving techniques. Another interesting topic is
the optimization of grounding in temporal ASP specifications as those handled
by telingo. The current grounding of telingo is inherited from incremental
solving in clingo and does not exploit the semantics of temporal expressions
that are available now in the input language. Finally, we envisage to extend the
telingo system with features of DEL (an extension to cope with dynamic logic
operators) in order to obtain a powerful system for representing and reasoning
about dynamic domains, not only providing an effective implementation of TEL
and DEL but, furthermore, a platform for action and control languages.

Herzig and Luis Fariñas del Cerro from the IRIT at Toulouse, among others.

References

1. Aguado, F., Cabalar, P., Diéguez, M., Pérez, G., Schaub, T., Schuhmann, A., Vidal,

Listing 1.2. telingo basic encoding for Moore’s problem

1 #const n=19.

2
3 #program initial.

4
5 process(a;b).

6 local (0;1).

7 instruction (0..2).

8
9 % Each process P executes:

10 % 0 assign c to r(P)

11 % 1 add c to r(P)

12 % 2 assing r(P) to C

13 %

14 % i(P)=I points to the next instruction I of process P to execute

15
16 holds(i(P),0) :- process(P).

17 holds(c,1).

18 holds(r(P),0) :- process(P).

19
20 #program dynamic.

21
22 %1 {fetch(P): _process(P)} 1.

23 {fetch(P,I): _instruction(I)} 1 :- _process(P).

24 fetch(P) :- fetch(P,I).

25 :- fetch(P,I), not _local(I), fetch(Q), P!=Q.

26 :- #count{P:fetch(P)}=0.

27
28 change(i(P),(I+1)\3) :- fetch(P), ’holds(i(P),I).

29 change(r(P),C ) :- fetch(P), ’holds(i(P),0), ’holds(c,C).

30 change(r(P),R+C) :- fetch(P), ’holds(i(P),1), ’holds(c,C),

31 ’holds(r(P),R), R+C <= n.

32 change(r(P),n+1) :- fetch(P), ’holds(i(P),1), ’holds(c,C),

33 ’holds(r(P),R), R+C > n.

34 change(c ,R ) :- fetch(P), ’holds(i(P),2), ’holds(r(P),R).

35
36 holds(F,V) :- change(F,V).

37 holds(F,V) :- ’holds(F,V), not change(F,_).

38
39 #program final.

40 :- not _testing , not holds(c,n).

41
42 #show fetch /1.

43 #show holds /2.

Listing 1.3. telingo second encoding for Moore’s problem

1 #const n=23.

2
3 #program initial.

4
5 process(a;b).

6 local (0;1).

7 instruction (0..2).

8
9 % Each process P executes:

10 % 0 assign c to r(P)

11 % 1 add c to r(P)

12 % 2 assing r(P) to C

13
14 holds(c,1).

15 holds(r(P),0) :- process(P).

16
17 #program dynamic.

18
19 {fetch(P,I): _instruction(I)} 1 :- _process(P).

20 fetch(P) :- fetch(P,I).

21 :- fetch(P,I), not _local(I), fetch(Q), P!=Q.

22 :- #count{P:fetch(P)}=0.

23
24
25 change(r(P),C ) :- fetch(P,0), ’holds(c,C).

26 change(r(P),R+C) :- fetch(P,1), ’holds(c,C), ’holds(r(P),R), R+C<=n.

27 change(r(P),n+1) :- fetch(P,1), ’holds(c,C), ’holds(r(P),R), R+C>n.

28 change(c ,R ) :- fetch(P,2), ’holds(r(P),R).

29
30 holds(F,V) :- change(F,V).

31 holds(F,V) :- ’holds(F,V), not change(F,_).

32
33 :- fetch(P,I), &tel{< <? fetch(P)},

34 not &tel { < (~ fetch(P) <? fetch(P,I’)) : I’=(I+2)\3 }.

35 :- fetch(P,I), not &tel{< <? fetch(P)}, I!=0.

36
37 #program final.

38 :- not _testing , not holds(c,n).

39
40 #show fetch /2.

41 #show holds /2.


